

NAG Library Routine Document

S30JBF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

S30JBF computes the European option price together with its sensitivities (Greeks) using the Merton jump-diffusion model.

2 Specification

```

SUBROUTINE S30JBF (CALPUT, M, N, X, S, T, SIGMA, R, LAMBDA, JVOL, P, LDP,      &
                  DELTA, GAMMA, VEGA, THETA, RHO, VANNA, CHARM, SPEED,      &
                  COLOUR, ZOMMA, VOMMA, IFAIL)
INTEGER           M, N, LDP, IFAIL
REAL (KIND=nag_wp) X(M), S, T(N), SIGMA, R, LAMBDA, JVOL, P(LDP,N),      &
                  DELTA(LDP,N), GAMMA(LDP,N), VEGA(LDP,N), THETA(LDP,N),  &
                  RHO(LDP,N), VANNA(LDP,N), CHARM(LDP,N), SPEED(LDP,N),  &
                  COLOUR(LDP,N), ZOMMA(LDP,N), VOMMA(LDP,N)
CHARACTER(1)     CALPUT

```

3 Description

S30JBF uses Merton's jump-diffusion model (Merton (1976)) to compute the price of a European option, together with the Greeks or sensitivities, which are the partial derivatives of the option price with respect to certain of the other input parameters. Merton's model assumes that the asset price is described by a Brownian motion with drift, as in the Black–Scholes–Merton case, together with a compound Poisson process to model the jumps. The corresponding stochastic differential equation is,

$$\frac{dS}{S} = (\alpha - \lambda k)dt + \hat{\sigma}dW_t + dq_t.$$

Here α is the instantaneous expected return on the asset price, S ; $\hat{\sigma}^2$ is the instantaneous variance of the return when the Poisson event does not occur; dW_t is a standard Brownian motion; q_t is the independent Poisson process and $k = E[Y - 1]$ where $Y - 1$ is the random variable change in the stock price if the Poisson event occurs and E is the expectation operator over the random variable Y .

This leads to the following price for a European option (see Haug (2007))

$$P_{\text{call}} = \sum_{j=0}^{\infty} \frac{e^{-\lambda T} (\lambda T)^j}{j!} C_j(S, X, T, r, \sigma'_j),$$

where T is the time to expiry; X is the strike price; r is the annual risk-free interest rate; $C_j(S, X, T, r, \sigma'_j)$ is the Black–Scholes–Merton option pricing formula for a European call (see S30AAF).

$$\sigma'_j = \sqrt{z^2 + \delta^2 \left(\frac{j}{T}\right)},$$

$$z^2 = \sigma^2 - \lambda \delta^2,$$

$$\delta^2 = \frac{\gamma \sigma^2}{\lambda},$$

where σ is the total volatility including jumps; λ is the expected number of jumps given as an average per year; γ is the proportion of the total volatility due to jumps.

The value of a put is obtained by substituting the Black–Scholes–Merton put price for $C_j(S, X, T, r, \sigma'_j)$.

4 References

Haug E G (2007) *The Complete Guide to Option Pricing Formulas* (2nd Edition) McGraw-Hill

Merton R C (1976) Option pricing when underlying stock returns are discontinuous *Journal of Financial Economics* 3 125–144

5 Parameters

- 1: CALPUT – CHARACTER(1) *Input*
On entry: determines whether the option is a call or a put.
 CALPUT = 'C'
 A call. The holder has a right to buy.
 CALPUT = 'P'
 A put. The holder has a right to sell.
Constraint: CALPUT = 'C' or 'P'.
- 2: M – INTEGER *Input*
On entry: the number of strike prices to be used.
Constraint: $M \geq 1$.
- 3: N – INTEGER *Input*
On entry: the number of times to expiry to be used.
Constraint: $N \geq 1$.
- 4: X(M) – REAL (KIND=nag_wp) array *Input*
On entry: X(*i*) must contain X_i , the *i*th strike price, for $i = 1, 2, \dots, M$.
Constraint: $X(i) \geq z$ and $X(i) \leq 1/z$, where $z = X02AMF()$, the safe range parameter, for $i = 1, 2, \dots, M$.
- 5: S – REAL (KIND=nag_wp) *Input*
On entry: S, the price of the underlying asset.
Constraint: $S \geq z$ and $S \leq 1.0/z$, where $z = X02AMF()$, the safe range parameter.
- 6: T(N) – REAL (KIND=nag_wp) array *Input*
On entry: T(*i*) must contain T_i , the *i*th time, in years, to expiry, for $i = 1, 2, \dots, N$.
Constraint: $T(i) \geq z$, where $z = X02AMF()$, the safe range parameter, for $i = 1, 2, \dots, N$.
- 7: SIGMA – REAL (KIND=nag_wp) *Input*
On entry: σ , the annual total volatility, including jumps.
Constraint: SIGMA > 0.0.
- 8: R – REAL (KIND=nag_wp) *Input*
On entry: r , the annual risk-free interest rate, continuously compounded. Note that a rate of 5% should be entered as 0.05.
Constraint: $R \geq 0.0$.

- 9: LAMBDA – REAL (KIND=nag_wp) Input
On entry: λ , the number of expected jumps per year.
Constraint: LAMBDA > 0.0.
- 10: JVOL – REAL (KIND=nag_wp) Input
On entry: the proportion of the total volatility associated with jumps.
Constraint: $0.0 \leq \text{JVOL} < 1.0$.
- 11: P(LDP,N) – REAL (KIND=nag_wp) array Output
On exit: the leading $M \times N$ part of the array P contains the computed option prices.
- 12: LDP – INTEGER Input
On entry: the first dimension of the arrays P, DELTA, GAMMA, VEGA, THETA, RHO, VANNA, CHARM, SPEED, COLOUR, ZOMMA and VOMMA as declared in the (sub)program from which S30JBF is called.
Constraint: LDP \geq M.
- 13: DELTA(LDP,N) – REAL (KIND=nag_wp) array Output
On exit: the leading $M \times N$ part of the array DELTA contains the sensitivity, $\frac{\partial P}{\partial S}$, of the option price to change in the price of the underlying asset.
- 14: GAMMA(LDP,N) – REAL (KIND=nag_wp) array Output
On exit: the leading $M \times N$ part of the array GAMMA contains the sensitivity, $\frac{\partial^2 P}{\partial S^2}$, of DELTA to change in the price of the underlying asset.
- 15: VEGA(LDP,N) – REAL (KIND=nag_wp) array Output
On exit: the leading $M \times N$ part of the array VEGA contains the sensitivity, $\frac{\partial P}{\partial \sigma}$, of the option price to change in the volatility of the underlying asset.
- 16: THETA(LDP,N) – REAL (KIND=nag_wp) array Output
On exit: the leading $M \times N$ part of the array THETA contains the sensitivity, $-\frac{\partial P}{\partial T}$, of the option price to change in the time to expiry of the option.
- 17: RHO(LDP,N) – REAL (KIND=nag_wp) array Output
On exit: the leading $M \times N$ part of the array RHO contains the sensitivity, $\frac{\partial P}{\partial r}$, of the option price to change in the annual risk-free interest rate.
- 18: VANNA(LDP,N) – REAL (KIND=nag_wp) array Output
On exit: the leading $M \times N$ part of the array VANNA contains the sensitivity, $\frac{\partial^2 P}{\partial S \partial \sigma}$, of VEGA to change in the price of the underlying asset or, equivalently, the sensitivity of DELTA to change in the volatility of the asset price.
- 19: CHARM(LDP,N) – REAL (KIND=nag_wp) array Output
On exit: the leading $M \times N$ part of the array CHARM contains the sensitivity, $-\frac{\partial^2 P}{\partial S \partial T}$, of DELTA to change in the time to expiry of the option.

- 20: SPEED(LDP,N) – REAL (KIND=nag_wp) array Output
On exit: the leading $M \times N$ part of the array SPEED contains the sensitivity, $\frac{\partial^3 P}{\partial S^3}$, of GAMMA to change in the price of the underlying asset.
- 21: COLOUR(LDP,N) – REAL (KIND=nag_wp) array Output
On exit: the leading $M \times N$ part of the array COLOUR contains the sensitivity, $-\frac{\partial^3 P}{\partial S^2 \partial T}$, of GAMMA to change in the time to expiry of the option.
- 22: ZOMMA(LDP,N) – REAL (KIND=nag_wp) array Output
On exit: the leading $M \times N$ part of the array ZOMMA contains the sensitivity, $\frac{\partial^3 P}{\partial S^2 \partial \sigma}$, of GAMMA to change in the volatility of the underlying asset.
- 23: VOMMA(LDP,N) – REAL (KIND=nag_wp) array Output
On exit: the leading $M \times N$ part of the array VOMMA contains the sensitivity, $\frac{\partial^2 P}{\partial \sigma^2}$, of VEGA to change in the volatility of the underlying asset.
- 24: IFAIL – INTEGER Input/Output
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**
On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, CALPUT \neq 'C' or 'P'.

IFAIL = 2

On entry, $M \leq 0$.

IFAIL = 3

On entry, $N \leq 0$.

IFAIL = 4

On entry, $X(i) < z$ or $X(i) > 1/z$, where $z = X02AMF()$, the safe range parameter.

IFAIL = 5

On entry, $S < z$ or $S > 1.0/z$, where $z = X02AMF()$, the safe range parameter.

IFAIL = 6

On entry, $T(i) < z$, where $z = X02AMF()$, the safe range parameter.

IFAIL = 7

On entry, $SIGMA \leq 0.0$.

IFAIL = 8

On entry, $R < 0.0$.

IFAIL = 9

On entry, $LAMBDA \leq 0.0$.

IFAIL = 10

On entry, $JVOL < 0.0$ or $JVOL \geq 1.0$.

IFAIL = 12

On entry, $LDP < M$.

7 Accuracy

The accuracy of the output is dependent on the accuracy of the cumulative Normal distribution function, Φ , occurring in C_j . This is evaluated using a rational Chebyshev expansion, chosen so that the maximum relative error in the expansion is of the order of the *machine precision* (see S15ABF and S15ADF). An accuracy close to *machine precision* can generally be expected.

8 Further Comments

None.

9 Example

This example computes the price of two European calls with jumps. The time to expiry is 6 months, the stock price is 100 and strike prices are 80 and 90 respectively. The number of jumps per year is 5 and the percentage of the total volatility due to jumps is 25%. The risk-free interest rate is 8% per year while the total volatility is 25% per year.

9.1 Program Text

```

Program s30jbf
!      S30JBF Example Program Text
!
!      Mark 24 Release. NAG Copyright 2012.
!
!      .. Use Statements ..
      Use nag_library, Only: nag_wp, s30jbf
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Real (Kind=nag_wp)         :: jvol, lambda, r, s, sigma
      Integer                    :: i, ifail, j, ldp, m, n
      Character (1)              :: calput
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: charm(:,,:), colour(:,,:), crho(:,,:), &
                                         delta(:,,:), gamma(:,,:), p(:,,:), &
                                         rho(:,,:), speed(:,,:), t(:), &

```

```

theta(:,:), vanna(:,:), vega(:,:), &
vomma(:,:), x(:), zomma(:,:)

! .. Executable Statements ..
Write (nout,*) 'S30JBF Example Program Results'

! Skip heading in data file
Read (nin,*)

Read (nin,*) calput
Read (nin,*) lambda
Read (nin,*) s, sigma, r, jvol
Read (nin,*) m, n

ldp = m
Allocate (charm(ldp,n), colour(ldp,n), crho(ldp,n), delta(ldp,n), &
gamma(ldp,n), p(ldp,n), rho(ldp,n), speed(ldp,n), t(n), theta(ldp,n), &
vanna(ldp,n), vega(ldp,n), vomma(ldp,n), x(m), zomma(ldp,n))

Read (nin,*)(x(i),i=1,m)
Read (nin,*)(t(i),i=1,n)

ifail = 0

Call s30jbf(calput,m,n,x,s,t,sigma,r,lambda,jvol,p,ldp,delta,gamma,vega, &
theta,rho,vanna,charm,speed,colour,zomma,vomma,ifail)

Write (nout,*)
Write (nout,*) 'Merton Jump-Diffusion Model'

Select Case (calput)
Case ('C','c')
Write (nout,*) 'European Call :'
Case ('P','p')
Write (nout,*) 'European Put :'
End Select

Write (nout,99996) ' Spot = ', s
Write (nout,99996) ' Volatility = ', sigma
Write (nout,99996) ' Rate = ', r
Write (nout,99996) ' Jumps = ', lambda
Write (nout,99996) ' Jump vol = ', jvol

Write (nout,*)

Do j = 1, n
Write (nout,*)
Write (nout,99999) t(j)
Write (nout,*) ' Strike Price Delta Gamma Vega Theta' &
// ' Rho'

Do i = 1, m
Write (nout,99998) x(i), p(i,j), delta(i,j), gamma(i,j), vega(i,j), &
theta(i,j), rho(i,j)
End Do

Write (nout,*) &
' Strike Price Vanna Charm Speed Colour Zomma' // &
' Vomma'

Do i = 1, m
Write (nout,99997) x(i), p(i,j), vanna(i,j), charm(i,j), speed(i,j), &
colour(i,j), zomma(i,j), vomma(i,j)
End Do

End Do

99999 Format (1X,'Time to Expiry : ',1X,F8.4)
99998 Format (1X,7(F8.4,1X))
99997 Format (1X,8(F8.4,1X))
99996 Format (A,1X,F8.4)
End Program s30jbf

```

9.2 Program Data

```

S30JBF Example Program Data
'C'           : Call = 'C', Put = 'P'
5.0          : LAMBDA (jumps)
100.0 0.25 0.08 0.25 : S, SIGMA, R, JVOL
2 1          : M, N
80.0
90.0          : X(I), I = 1,2,...M
0.5          : T(I), I = 1,2,...N

```

9.3 Program Results

S30JBF Example Program Results

Merton Jump-Diffusion Model

European Call :

```

Spot          = 100.0000
Volatility    = 0.2500
Rate          = 0.0800
Jumps        = 5.0000
Jump vol     = 0.2500

```

Time to Expiry : 0.5000

Strike	Price	Delta	Gamma	Vega	Theta	Rho	
80.0000	23.6090	0.9431	0.0064	8.1206	-7.6718	35.3480	
90.0000	15.4193	0.8203	0.0149	18.5256	-9.9695	33.3037	
Strike	Price	Vanna	Charm	Speed	Colour	Zomma	Vomma
80.0000	23.6090	-0.6334	0.1080	-0.0006	-0.0035	0.0315	70.6824
90.0000	15.4193	-0.7726	0.0770	-0.0009	0.0109	-0.0186	49.7161
